

# NAG Fortran Library Routine Document

## F03AAF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

F03AAF calculates the determinant of a real matrix using an *LU* factorization with partial pivoting.

### 2 Specification

```
SUBROUTINE F03AAF(A, IA, N, DET, WKSPCE, IFAIL)
INTEGER          IA, N, IFAIL
real           A(IA,*), DET, WKSPCE(*)
```

### 3 Description

This routine calculates the determinant of *A* using the *LU* factorization with partial pivoting,  $PA = LU$ , where *P* is a permutation matrix, *L* is lower triangular and *U* is unit upper triangular. The determinant of *A* is the product of the diagonal elements of *L* with the correct sign determined by the row interchanges.

### 4 References

Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer-Verlag

### 5 Parameters

- 1: *A*(IA,\*) – **real** array *Input/Output*  
**Note:** the second dimension of the array *A* must be at least  $\max(1, N)$ .  
*On entry:* the *n* by *n* matrix *A*.  
*On exit:* *A* is overwritten by the factors *L* and *U*, except that the unit diagonal elements of *U* are not stored.
- 2: IA – INTEGER *Input*  
*On entry:* the first dimension of the array *A* as declared in the (sub)program from which F03AAF is called.  
*Constraint:*  $IA \geq \max(1, N)$ .
- 3: N – INTEGER *Input*  
*On entry:* *n*, the order of the matrix *A*.  
*Constraint:*  $N \geq 0$ .
- 4: DET – **real** *Output*  
*On exit:* the determinant of *A*.
- 5: WKSPCE(\*) – **real** array *Workspace*  
**Note:** the dimension of the array WKSPCE must be at least  $\max(1, N)$ .

## 6: IFAIL – INTEGER

*Input/Output*

*On entry:* IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

The matrix  $A$  is singular, possibly due to rounding errors. The factorization could not be completed. DET is set to 0.0.

IFAIL = 2

Overflow. The value of the determinant is too large to be held in the computer.

IFAIL = 3

Underflow. The value of the determinant is too small to be held in the computer.

IFAIL = 4

On entry,  $N < 0$ ,  
or  $IA < \max(1, N)$ .

## 7 Accuracy

The accuracy of the determinant depends on the conditioning of the original matrix. For a detailed error analysis see page 107 of Wilkinson and Reinsch (1971).

## 8 Further Comments

The time taken by the routine is approximately proportional to  $n^3$ .

## 9 Example

To calculate the determinant of the real matrix

$$\begin{pmatrix} 33 & 16 & 72 \\ -24 & -10 & -57 \\ -8 & -4 & -17 \end{pmatrix}.$$

## 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F03AAF Example Program Text
*      Mark 15 Revised.  NAG Copyright 1991.
*      .. Parameters ..
      INTEGER          NMAX, IA
      PARAMETER        (NMAX=8,IA=NMAX)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*      .. Local Scalars ..
      real             DETERM
      INTEGER          I, IFAIL, J, N
*      .. Local Arrays ..
      real             A(IA,NMAX), WKSPCE(NMAX)
*      .. External Subroutines ..
      EXTERNAL         F03AAF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F03AAF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      WRITE (NOUT,*)
      IF (N.GE.0 .AND. N.LE.NMAX) THEN
         READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
         IFAIL = 0
*
         CALL F03AAF(A,IA,N,DETERM,WKSPCE,IFAIL)
*
         WRITE (NOUT,99998) 'Value of determinant = ', DETERM
      ELSE
         WRITE (NOUT,99999) 'N is out of range: N = ', N
      END IF
      STOP
*
99999  FORMAT (1X,A,I5)
99998  FORMAT (1X,A,F9.4)
      END
```

## 9.2 Program Data

```
F03AAF Example Program Data
3
 33   16   72
-24  -10  -57
 -8   -4  -17
```

## 9.3 Program Results

F03AAF Example Program Results

Value of determinant =     6.0000

---